NASA Technical Memorandum 74086

(NASA-TM-74086) PRELIMINARY STUDY OF THE USE OF THE STAR-100 COMPUTER FOR TRANSONIC FLOW CALCULATIONS (NASA) 21 p HC A02/MF A01 CSCL 01A

N78-11005

Unclas
G3/02 52814

PRELIMINARY STUDY OF THE USE OF THE STAR-100 COMPUTER FOR TRANSONIC FLOW CALCULATIONS

JAMES D. KELLER AND ANTONY JAMESON

NOVEMBER 1977

NASA
National Aeronautics and Space Administration

Langley Research Center
Hampton, Virginia 23665

PRELIMINARY STUDY OF THE USE OF THE STAR-100 COMPUTER FOR TRANSONIC FLOW CALCULATIONS

James D. Keller and Antony Jameson*
Langley Research Center

SUMMARY

A new explicit method for solving the transonic small-disturbance potential equation is presented. This algorithm, which is suitable for the new vector-processor computers such as the CDC STAR-100, is compared to successive line over-relaxation (SLOR) on a simple test problem. The convergence rate of the explicit scheme is slower than that of SLOR. However, the efficiency of the explicit scheme on the STAR-100 computer is sufficient to overcome the slower convergence rate and allow an overall speedup compared to SLOR on the CYBER 175 computer.

INTRODUCTION

The state-of-the-art of transonic flow calculations has advanced to the point where two-dimensional flows, including the effects of viscosity, can be computed in a relatively short time on modern serial-type computers. For example, many people are using a program developed at the Courant Institute of New York

---

*Courant Institute of Mathematical Sciences, New York University

University for the analysis of transonic flow past airfoils (ref. 1). This program gives accurate solutions to the full-potential equation, including the effects of boundary-layer displacement, in about 2 or 3 minutes on a CDC CYBER 175 computer. Three-dimensional, transonic, finite-difference calculations, however, are expensive on this type of computer. The three-dimensional program described in reference 1 takes about half an hour for an inviscid calculation on a fairly crude grid.

It is hoped that the use of the STAR-100 computer will allow accurate, three-dimensional, transonic flow calculations to be done economically. One way to achieve this goal is through the development of algorithms which can make full use of the unique architecture of the STAR-100. The STAR computer has a "pipeline" type of processor which is very efficient in doing arithmetic operations on long vectors (ref. 2). Unfortunately, the best available method for solving the transonic potential equation is successive line over-relaxation (SLOR), which is not amenable to vector arithmetic. The reason for this is the semi-implicit nature of the iterative method; that is, the calculations at a particular grid point require results from the current iteration at neighboring grid points and thus cannot be done in long vector operations.

This paper describes a new explicit algorithm which can be vectorized for use on the STAR-100. The new algorithm is applied to a simple test case and compared to SLOR on the CYBER 175

computer. A simple STAR program, including vector instructions, is given in the appendix.

TEST PROBLEM

The test problem chosen for this preliminary study is to solve the transonic, nonlinear, small-disturbance potential equation for a nonlifting parabolic-arc airfoil in a finite box with uniform grid, as shown in the sketch below.



Although this is a very simple physical situation, it still has some of the most difficult features of transonic flow fields as far as programing for the STAR is concerned. The governing partial differential equation is

$$\left[1 - M_\infty^2 - (\gamma+1)\, M_\infty^2\, \phi_x\right]\phi_{xx} + \phi_{yy} = 0$$

The boundary conditions to be applied are

$\phi = 0$ on the outer boundary

and

$$\phi_y = \mp 4\tau x \quad \text{on} \quad y = \pm 0, \; -.5 \leq x \leq .5$$

where τ is the thickness of the parabolic-arc airfoil (which has a unit chord). For $\tau = 0.1$ and $M_\infty = 0.9$ the flow is supercritical. In regions where the coefficient of ϕ_{xx} is positive, the flow is subsonic and the equation is elliptic type. In regions where the coefficient of ϕ_{xx} is negative, the flow is supersonic and the equation is hyperbolic type. The general procedure for solving this equation is to replace the partial differential equation with a finite difference equation at each grid point. These finite difference equations are then solved iteratively.

This test problem represents a simple physical situation which is of little practical interest. A more useful program should allow for lifting flows and should extend the outer boundary farther away from the airfoil. This could be done either by using a stretched grid or by some type of grid nesting using additional coarse grids around the small region considered here. The test case does, however, include the major difficulties to be overcome in using the STAR-100 computer for transonic flows. For example, it has supercritical flow which requires a change from one type of difference equation at subsonic (elliptic) points to another type of difference equation at supersonic (hyperbolic) points. It also requires the use of an explicit iterative scheme to solve the

difference equations if the arithmetic operations are to be done using long vector instructions.

SEMI-IMPLICIT SOLUTION METHOD

The most common method used to solve the finite difference equations is successive line over-relaxation (SLOR). This iterative scheme is implemented as follows:

Compute $U \equiv 1 - M_\infty^2 - (\gamma+1)\, M_\infty^2\, \phi_x$

using

$$\phi_x = \frac{\phi^n_{i+1,j} - \phi^n_{i-1,j}}{2\Delta x}$$

If $U > 0$ (subsonic points), central differences are used together with over-relaxation ($\omega > 1$) to give:

$$U\,\frac{\phi^n_{i+1,j} - \frac{2}{\omega}\,\phi^{n+1}_{i,j} - 2\left(1 - \frac{1}{\omega}\right)\phi^n_{i,j} + \phi^{n+1}_{i-1,j}}{\Delta x^2}$$

$$+\,\frac{\phi^{n+1}_{i,j+1} - 2\phi^{n+1}_{i,j} + \phi^{n+1}_{i,j-1}}{\Delta y^2} = 0$$

where the superscript indicates the iteration number.

If $U < 0$ (supersonic points), an upwind difference is used for ϕ_{xx} to give:

$$U\,\frac{\phi^{n+1}_{i,j} - 2\phi^{n+1}_{i-1,j} + \phi^{n+1}_{i-2,j}}{\Delta x^2} + \frac{\phi^{n+1}_{i,j+1} - 2\phi^{n+1}_{i,j} + \phi^{n+1}_{i,j-1}}{\Delta y^2} = 0$$

At the airfoil boundary the ϕ_{yy} term is replaced by

$$\phi_{yy} = \frac{\phi_{i,j+1} - 2\phi_{i,j} + \phi_{i,j-1}}{\Delta y^2} - \frac{2\phi_y\big|_{y=0}}{\Delta y}$$

which takes into account the fact that the nonlifting flow is symmetric (i.e., $\phi_{i,j-1} = \phi_{i,j+1}$ at $y = 0$). Each iteration is generated one column at a time going from left to right, by solving a tridiagonal system of equations at each column. This nonconservative scheme is similar to that originally proposed by Murman and Cole (ref. 3). See references 4 and 5 for a discussion of related conservative schemes.

EXPLICIT SOLUTION METHOD

The explicit solution method uses values of the potential function from the two previous iterations in order to update the potential function for the present iteration (thus termed a three-level scheme). The method is implemented as follows:

Compute $U \equiv 1 - M_\infty^2 - (\gamma+1)\, M_\infty^2\, \phi_x$

using

$$\phi_x = \frac{\phi^n_{i+1,j} - \phi^n_{i-1,j}}{2\Delta x}$$

If $U > 0$ (subsonic points), central differences are used at iteration, or "time," level n:

$$\phi_{i,j}^{n+1} - 2\phi_{i,j}^{n} + \phi_{i,j}^{n-1} + (2 - P_1)\left(\phi_{i,j}^{n} - \phi_{i,j}^{n-1}\right) = D_1 R_1$$

where

$$R_1 = U\left(\phi_{i+1,j}^{n} - 2\phi_{i,j}^{n} + \phi_{i-1,j}^{n}\right) + \frac{\Delta x^2}{\Delta y^2}\left(\phi_{i,j+1}^{n} - 2\phi_{i,j}^{n} + \phi_{i,j-1}^{n}\right)$$

and

$$D_1 = \frac{P_1 P_2^{\,2}}{2\left[U + \left(\frac{\Delta x}{\Delta y}\right)^2\right]}$$

If $U < 0$ (supersonic points), the upwind ϕ_{xx} difference is formed at "time" level $n - 1$, while ϕ_{yy} is centrally-differenced and averaged in time level n, to give:

$$\phi_{i,j}^{n+1} - 2\phi_{i,j}^{n} + \phi_{i,j}^{n-1} + 2\sigma\left(\phi_{i,j}^{n} - \phi_{i,j}^{n-1} - \phi_{i-1,j}^{n} + \phi_{i-1,j}^{n-1}\right)$$

$$= D_2 R_2 \qquad (1)$$

where

$$R_2 = \overline{U}\left(\phi^{n-1}_{i,j} - 2\phi^{n-1}_{i-1,j} + \phi^{n-1}_{i-2,j}\right)$$

$$+ \left(\frac{\Delta x}{\Delta y}\right)^2 \left[(1-\sigma)\left(\phi^{n}_{i,j+1} - 2\phi^{n}_{i,j} + \phi^{n}_{i,j-1}\right)\right.$$

$$\left. + \sigma\left(\phi^{n}_{i-1,j+1} - 2\phi^{n}_{i-1,j} + \phi^{n}_{i-1,j-1}\right)\right]$$

$$\overline{U} = (1-\sigma)\, U + \sigma\widetilde{U}$$

$$\widetilde{U} = 1 - M_\infty^2 - (\gamma+1)\, M_\infty^2 \left(\frac{\phi^{n-1}_{i,j} - \phi^{n-1}_{i-2,j}}{2\Delta x}\right)$$

$$\sigma = P_2 \min\left(1, \sqrt{|U|}\,\frac{\Delta y}{\Delta x}\right)$$

and

$$D_2 = \frac{P_2{}^2\left(\frac{\Delta y}{\Delta x}\right)^2}{\max\left[1, |U|\left(\frac{\Delta y}{\Delta x}\right)^2\right]}$$

A von Neumann stability analysis of this last scheme with the $\overline{U}$ replaced by U shows why the ϕ_{xx} derivative is evaluated

at the n-1 iteration and why the ϕ_{yy} derivative is a weighted average between ϕ_{yy} at the i column and ϕ_{yy} at the i-1 column. To do the von Neumann analysis, let ϕ at the kth iteration be

$$\phi^{(k)} = g^k e^{imx} e^{iny}$$

Also let $m\Delta x = \xi$ and $n\Delta y = \eta$. Putting these definitions into equation (1) gives

$$g^2 - 2g + 1 + 2\sigma (g - 1 - ge^{-i\xi} + e^{-i\xi})$$

$$= D_2 \left\{ U (1 - 2e^{-i\xi} + e^{-2i\xi}) \right.$$

$$+ g \left(\frac{\Delta x}{\Delta y}\right)^2 \left[(1-\sigma)(e^{i\eta} - 2 + e^{-i\eta}) \right.$$

$$\left. \left. + \sigma (e^{-i\xi} e^{i\eta} - 2e^{-i\xi} e^{-i\eta}) \right] \right\} \qquad (2)$$

Note that

$$D_2 U = - \sigma^2$$

Also define

$$\rho \equiv 1 - 2\sigma^2 \frac{1}{|U|} \left(\frac{\Delta x}{\Delta y}\right)^2 \sin^2 \frac{\eta}{2}$$

so equation (2) becomes

$$g^2 - 2\rho g\left[1 - \sigma\,(1 - e^{-i\xi})\right] + \left[1 - \sigma\,(1 - e^{i\xi})\right]^2 = 0$$

or

$$g = \left(\rho \pm i\sqrt{1 - \rho^2}\right)\left|1 - \sigma\,(1 - e^{-i\xi})\right|$$

Thus, in order to have $|g| \leq 1$, it is necessary and sufficient to have both $|\rho| \leq 1$ and $\sigma \leq 1$. The inequality $|\rho| \leq 1$ implies that $\sigma \leq \sqrt{|U|}\,\frac{\Delta y}{\Delta x}$.

This convenient factoring of the expression for the amplification factor was made possible by choosing this particular representation for the ϕ_{xx} and ϕ_{yy} derivatives. In practice the coefficient of ϕ_{xx} is $\bar{U}$ rather than U; this averaging makes the scheme approach second-order accuracy as σ approaches one.

It should be noted that this explicit scheme is not only a different iterative algorithm from the semi-implicit scheme, but it also has a different steady-state solution. This difference occurs because of the weighted averaging done on both the ϕ_{yy} term and on the coefficient of the ϕ_{xx} term for supersonic points.

CONVERGENCE AND TIMING COMPARISONS

Short computer programs have been written to solve the sample problem using the two methods described. The semi-implicit SLOR

method was coded in standard FORTRAN IV and run on a CDC CYBER 175 computer (which is about $2\frac{1}{2}$ times as fast as a CDC 6600 for this type of problem). The explicit method was coded in STAR FORTRAN (ref. 6) and run on a CDC STAR-100 computer. The STAR code includes the use of vector instructions in the iteration loop. It also includes the use of bit control vectors to distinguish between subsonic and supersonic points. The bit control vectors provide the capability of performing the complicated supersonic calculations only at supersonic points, which are collected into a vector through the "compress" and "expand" type of instructions available on the STAR. The STAR code for this problem is listed in the appendix.

The computations were done on three different grids. Each calculation was terminated when the value of the largest residual in the flow field was less than $\frac{1}{2}(\Delta x^2 + \Delta y^2)$. Each calculation was run with experimentally-determined optimum values of the parameters for that algorithm so that convergence was attained in a minimum number of cycles. The results in the table below show that the new three-level explicit scheme has a slower convergence

Grid size	SLOR ON CYBER 175			EXPLICIT METHOD ON STAR-100		
	Cycles to converge	Time to converge	Average sec/cy.	Cycles to converge	Time to converge	Average sec/cy.
40X40	42	1.242	.0296	131	.879	.0067
80X80	98	12.118	.1237	300	6.385	.0213
160X160	244	116.095	.4758	655	59.834	.0913

rate than SLOR. However, the efficiency of this new scheme on the STAR computer is enough to make up for the slower convergence rate and still allow an overall reduction in computing time. Not surprisingly, the speedup is greater for the cases with finer grids than for the 40X40 case.

CONCLUDING REMARKS

This preliminary study has shown that a new explicit method for solving the transonic small-disturbance potential equation on the STAR-100 computer can almost halve the computer time required for this type of computation when compared to successive line over-relaxation on the CDC CYBER 175 computer. These results are limited to a relatively simple problem with a uniform cartesian grid. Although the speedup is not as great as desired, it is enough to justify further study of this method. The effects of lift and of grid stretching on the convergence rates of the schemes should be investigated. Also, the new explicit scheme should be applied to the full potential equation.

There are several possibilities for obtaining further reductions in computer time. One is through the development of more efficient algorithms. Improvements might be made in the convergence rate of explicit algorithms, or other vectorizable algorithms might be developed. Another possibility is through programing techniques to get successive line over-relaxation to run as efficiently as possible on the STAR-100 computer. Although

the method is semi-implicit, there are portions of it which can be written in vector instructions of short length.

The program listing in the appendix should serve as an introduction to some of the programing techniques available on the STAR-100 which are useful for transonic flow calculations.

APPENDIX

PROGRAM LISTING

```
      PROGRAM UT(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      COMMON P(41,41),FP(41),IMAX,JMAX,JW,COVR,MIT,IMAXM1,JMAXM1
     * ,DX,DY,DX2R,P1,P2,XMINF,RSH,DY2R
      CALL R3CLOCKS(CPU,WALL)
      WRITE(6,9902)CPU,WALL
      XMINF=.9
      P1=1.83
      P2=.99
      RSH=1.4
      IMAX=41
      JMAX=41
      MIT=1000
      CF=1.
      PW=4./3.
      PH=4./3.
      IMAXM1=IMAX-1
      JMAXM1=JMAX-1
      DX=PW/IMAXM1
      DY=PH/JMAXM1
      JH=JMAXM1/2+1
      DX2R=1./DX**2
      DY2R=1./DY**2
      COVR=CF*.5*(DX**2+DY**2)
      DO 10 I=1,IMAX
      FP(I)=0.
   10 CONTINUE
      XLE=-.5
      XTE=.5
      EPS=1.0E-06
      IH=IMAXM1/2+1
      ILE=IFIX((XLE-EPS)/DX)+IH
      ITE=IFIX((XTE+EPS)/DX)+IH
      DO 11 I=ILE,ITE
      X=(I-IH)*DX
      FP(I)=.4*X
   11 CONTINUE
      NT=IMAX*JMAX
      P(1,1,NT)=0.
      CALL PICTURE
      CALL PRESS
      STOP
 9902 FORMAT(1X,A10,1X,A10)
      END
```

```
      SUBROUTINE PICTURE
C     THIS SUBROUTINE DOES THE RELAXATION
      COMMON P(41,41),PP(41),IMAX,JMAX,JW,COVR,MIT,IMAXM1,JMAXM1
     * ,DX,DY,DX2R,P1,P2,XMINF,RSH,DY2R
C     DIMENSION SOME ARRAYS FOR TEMPORARY STORAGE
      DIMENSION PP(41,41),PM(41,41)
      DIMENSION T1(41,41),T2(41,41),T3(41,41)
      BIT MU(41,41),ER(41,41),IB(41,41),MUD,EBD,IBD,B1D
      DESCRIPTOR T1D,T2D,T3D,PD,PPD,PMD
      DESCRIPTOR MUD,EBD,IBD,B1D
      DESCRIPTOR T1ND,T2ND,T3ND,T4ND,T5ND,T6ND,T7ND,T8ND
      CALL D3CLOCKS(CPU,WALL)
      WRITE(6,9904)CPU,WALL
C     ASSIGN THE DESCRIPTORS TO SPECIFIC ARRAYS
      NP=IMAX*JMAX-2*JMAX-2
      ASSIGN T1D,T1(2,2;NP)
      ASSIGN T2D,T2(2,2;NP)
      ASSIGN T3D,T3(2,2;NP)
      ASSIGN PD,P(2,2;NP)
      ASSIGN PPD,PP(2,2;NP)
      ASSIGN PMD,PM(2,2;NP)
      ASSIGN MUD,MU(2,2;NP)
      ASSIGN EBD,EB(2,2;NP)
      ASSIGN IBD,IB(2,2;NP)
C     SET UP A BIT ARRAY THAT HAS ONES AT INTERIOR POINTS
C     AND ZEROS AT THE EDGE POINTS
      IBD=B'0'
      DO 1 I=2,IMAXM1
      DO 1 J=2,JMAXM1
    1 IB(I,J)=B'1'
      EBD=.NOT.IBD
      XK=1.-XMINF**2
      AC=(RSH+1.)*XMINF**2/(2.*DX)
      D1=0.5*P1*P2*P2
      AR2=(DX/DY)**2
      AR2R=1./AR2
      D2=P2*P2/AR2
      TWODY=2.*DY
      S=P1-1.
C     INITIALIZE THE TEMPORARY ARRAYS
      T1D=0.
      T2D=0.
      T3D=0.
      PPD=0.
      PMD=0.
      WRITE(6,9901)
      POINTS=(IMAX-2)*(JMAX-2)
      RMAXO=10.
      RAVGO=10.
      RASUBO=10.
```

```
      RASUPO=10.
      N=1
   50 CONTINUE
C    COMPUTE U
      T1D=XK-AC*(P(3,2,NP)-P(1,2,NP))
C    SET THE MU(I,J) BIT TO ONE AT SUPERSONIC POINTS
      MUD=T1D.LT.0.
C    SET MU TO ZERO AT THE EDGE POINTS
      MUD=MUD.AND.IBD
C    COMPUTE A CENTRAL PHI-XX AT I,J
      PPD=P(1,2,NP)-PD-PD+P(3,2,NP)
C    COMPUTE A CENTRAL PHI-YY AT I,J
      T2D=P(2,1,NP)-PD-PD+P(2,3,NP)
C    COMPUTE THE RESIDUAL
      PPD=T1D*PPD+AR2*T2D
C    CHANGE THE RESIDUAL AT THE AIRFOIL SURFACE
      PP(2,JM,IMAX-2)=PP(2,JM,IMAX-2)+TWODY*FP(2,IMAX-2)
      T3D=0.
C    PUT ZEROS IN THE RESIDUAL ARRAY AT THE EDGE POINTS
      PPD=G8VCTRL(T3D,EBD,PPD)
C    PUT ZEROS IN THE RESIDUAL ARRAY AT SUPERSONIC POINTS
      PPD=G8VCTRL(T3D,MUD,PPD)
      T3D=VABS(PPD,T3D)
      PSUMSUB=G8SSUM(T3D)*DX2R
      RMAXSUB=G8SMAX(T3D)*DX2R
C    COMPUTE THE NEW PHI VALUES
      PPD=D1/(T1D+AR2)*PPD+PD-(1.-P1)*(PD-PMD)
C    NUMBER OF SUPERSONIC POINTS
      NS=G8SCNT(MUD)
      RSUMSUP=0.
      RMAXSUP=0.
      IF(NS.EQ.0)GO TO 300
C    SUPERSONIC CALCULATIONS
C    ASSIGN TEMPORARY STORAGE FOR SUPERSONIC CALCULATIONS
      ASSIGN T1ND,,DYN,NS
      ASSIGN T2ND,,DYN,NS
      ASSIGN T3ND,,DYN,NS
      ASSIGN T4ND,,DYN,NS
      ASSIGN T5ND,,DYN,NS
      ASSIGN T6ND,,DYN,NS
      ASSIGN T7ND,,DYN,NS
      ASSIGN T8ND,,DYN,NS
      ASSIGN B1D,,DYN,NS
C    MAKE A VECTOR OUT OF THE VALUES OF U AT SUPERSONIC POINTS
      T1ND=G8VCMPRS(T1D,MUD,T1ND)
C    COMPUTE SIGMA
      T3ND=VABS(T1ND,T3ND)*AR2R
      T4ND=VSQRT(T3ND,T4ND)
      B1D=T4ND.GT.1.
      T5ND=1.
```

```
      T4ND=P2*Q8VCTRL(T5ND,B1D,T4ND)
C     MAKE A VECTOR OUT OF THE VALUES OF THE CENTRAL
C     PHI-YY AT SUPERSONIC POINTS
      T2ND=Q8VCMPRS(T2D,MUD,T2ND)
C     MAKE A VECTOR OUT OF THE VALUES OF THE CENTRAL
C     PHI-YY ONE POINT UPSTREAM OF EACH SUPERSONIC POINT
      T5ND=Q8VCMPRS(T2(1,2,NP),MUD,T5ND)
C     COMPUTE PHI-YY TO USE AT SUPERSONIC POINTS
      T5ND=T2ND+T4ND*(T5ND-T2ND)
C     MAKE VECTORS OUT OF OLD VALUES OF PHI AT SUPERSONIC POINTS
C     AND ONE AND TWO POINTS UPSTREAM OF EACH SUPERSONIC POINT
      T6ND=Q8VCMPRS(PHD,MUD,T6ND)
      T7ND=Q8VCMPRS(PM(1,2,NP),MUD,T7ND)
      T8ND=Q8VCMPRS(PM(0,2,NP),MUD,T8ND)
C     COMPUTE U-TILDE
      T2ND=XK-AC*(T6ND-T8ND)
C     COMPUTE U-BAR
      T2ND=T1ND-T4ND*(T1ND-T2ND)
C     COMPUTE THE RESIDUAL
      T5ND=T5ND+T2ND*(T6ND-T7ND-T7ND+T8ND)
C     EXPAND THE RESIDUALS AT SUPERSONIC POINTS BACK
C     TO A TEMPORARY FULL ARRAY
      T3D=Q8VXPND(T5ND,MUD,T3D)
C     CHANGE THE RESIDUALS AT THE AIRFOIL SURFACE
      T3(2,JW,IMAX-2)=T3(2,JW,IMAX-2)+TWODY*FP(2,IMAX-2)
C     RECOMPRESS THE RESIDUALS AT SUPERSONIC POINTS
      T5ND=Q8VCMPRS(T3D,MUD,T5ND)
C     COMPUTE D2
      T1ND=VABS(T5ND,T1ND)
      RSUMSUP=Q8SSUM(T1ND)*DX2R
      RMAXSUP=Q8SMAX(T1ND)*DX2R
      T1ND=1.
      T1ND=D2/Q8VCTRL(T3ND,B1D,T1ND)
C     COMPUTE THE NEW VALUES OF PHI AT SUPERSONIC POINTS
      T8ND=Q8VCMPRS(PD,MUD,T8ND)
      T2ND=T8ND-T6ND
      T3ND=Q8VCMPRS(P(1,2,NP),MUD,T3ND)
      T3ND=T3ND-T7ND
      T5ND=T1ND*T5ND+T8ND+T2ND-(T4ND+T4ND)*(T2ND-T3ND)
C     EXPAND THE NEW VALUES OF PHI AT SUPERSONIC POINTS TO A FULL
C     ARRAY AND PUT THEM IN THE PHI-PLUS ARRAY
      T3D=Q8VXPND(T5ND,MUD,T3D)
      PPD=Q8VCTRL(T3D,MUD,PPD)
      FREE
  300 CONTINUE
C     MOVE THE PHI ARRAYS TO NEW ITERATION LEVEL
      PHD=PD
      PD=PPD
      RSUM=RSUMSUB+RSUMSUP
      RMAX=AMAX1(RMAXSUB,RMAXSUP)
```

```
      NTEMP=NS
      IF(NS.EQ.0)NTEMP=1
      FNS=FLOAT(NTEMP)
      RASUP=RSUMSUP/FNS
      IF(NS.EQ.0)RASUP=1.0
      RASUB=(RSUM-RSUMSUP)/(POINTS-FNS)
      RAVG=RSUM/POINTS
      SRM=RMAX/RMAXO
      SRA=RAVG/RAVGO
      SRASUB=RASUB/RASUBO
      SRASUP=RASUP/RASUPO
      RMAXO=RMAX
      RAVGO=RAVG
      RASUPO=RASUB
      RASUBO=RASUP
      WRITE(6,9902)N,RMAX,SRM,RAVG,SRA,RASUB,SRASUB,PASUP,SRASUP,NS
      IF(RMAX.LE.COVR.OR.N.GE.MIT)GO TO 500
      IF(N.EQ.1)RMAXI=RMAX
      IF(N.EQ.1)RAVGI=RAVG
      N=N+1
      GO TO 50
  500 CONTINUE
      SRM=(RMAX/RMAXI)**(1./FLOAT(N))
      SRA=(RAVG/RAVGI)**(1./FLOAT(N))
      WRITE(6,9903)SRM,SRA
      CALL R3CLOCKS(CPU,WALL)
      WRITE(6,9905)CPU,WALL
      RETURN
 9901 FORMAT(3H0 N4X4HRMAX6X3HSRM5X4HRAVG6X3HSRA4X5HRASUB5X
     * 6HSRASUB2X5HRASUP5X5HSRSUP3X2HNS)
 9902 FORMAT(1X,I3,E11.5,F7.4,E11.5,F7.4,E11.5,F7.4,E11.5,F7.4,I4)
 9903 FORMAT('0OVERALL SPECTRAL RADIUS IS'F7.4,' BASED ON RMAX AND'
     * F7.4,' BASED ON RAVG.')
 9904 FORMAT('0ENTERING PICTURE AFTER'F10.5,' SECONDS CPU TIME AND'
     * F10.5,' SECONDS ELAPSED TIME')
 9905 FORMAT(' ITERATIONS TOOK'F10.5,' SECONDS CPU TIME AND'F10.5,
     * ' SECONDS ELAPSED TIME')
      END
      SUBROUTINE PRESS
      COMMON P(41,41),FP(41),IMAX,JMAX,JW,COVR,MIT,IMAXM1,JMAXM1
     * ,DX,DY,DX2R,P1,P2,XMINF,RSH,DY2R
      DIMENSION CP(41)
      J=JW
      DO 1 I=2,IMAXM1
      CP(I)=(P(I-1,J)-P(I+1,J))/DX
    1 CONTINUE
      WRITE(6,9902)(I,CP(I),I=2,IMAXM1)
      RETURN
 9902 FORMAT('1   I    CP'/(I4,F10.6))
      END
```

REFERENCES

1. Bauer, Frances; Garabedian, Paul; Korn, David; and Jameson, Antony: Supercritical Wing Sections II. Lecture Notes in Economics and Mathematical Systems, vol. 108, Springer-Verlag, 1975.

2. Control Data STAR-100 Computer Hardware Reference Manual. CDC Publication No. 60256000, 1975.

3. Murman, Earll M.; and Cole, Julian D.: Calculation of Plane Steady Transonic Flows. AIAA J., vol. 9, no. 1, 1971, pp. 114-121.

4. Murman, Earll M.: Analysis of Embedded Shock Waves Calculated by Relaxation Methods. AIAA Computational Fluid Dynamics Conference, Palm Springs, Calif., 1973.

5. Jameson, Antony: Transonic Potential Flow Calculations Using Conservation Form. AIAA 2nd Computational Fluid Dynamics Conference, Hartford, Conn., 1975.

6. STAR FORTRAN Language Version 2 Reference Manual. CDC Publication No. 60386200, 1977.

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No
NASA TM 74086		

4. Title and Subtitle	5. Report Date
Preliminary Study of the Use of the STAR-100 Computer for Transonic Flow Calculations	November 1977
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
James D. Keller and Antony Jameson	
	10. Work Unit No.
9. Performing Organization Name and Address	505-06-13-01
NASA Langley Research Center Hampton, Virginia 23665	11. Contract or Grant No
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	NASA Technical Memorandum
National Aeronautics and Space Administration Washington, DC 20546	14. Sponsoring Agency Code

15. Supplementary Notes

Antony Jameson - Courant Institute of Mathematical Sciences, New York University

16. Abstract

A new explicit method for solving the transonic small-disturbance potential equation is presented. This algorithm, which is suitable for the new vector-processor computers such as the CDC STAR-100, is compared to successive line over-relaxation (SLOR) on a simple test problem. The convergence rate of the explicit scheme is slower than that of SLOR. However, the efficiency of the explicit scheme on the STAR-100 computer is sufficient to overcome the slower convergence rate and allow an overall speedup compared to SLOR on the CYBER 175 computer.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Transonic Flow Vector Computers Computer Programing and Software Aerodynamics	Unclassified - Unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	19	$3.50

* For sale by the National Technical Information Service, Springfield, Virginia 22161